
1 Introduction

The role of the following manual is not to formally de�ne the working and the
interpretations of the skeletal semantics. If that is what they are looking for,
the reader should refer to Victoire Noizet's thesis dissertation (or something
more recent if available). The goal of this manual is to provide a begginer's
guide to understand and handle skeletal semantics in the Necro ecosystem.
It is purposely simpli�ed, and limits itself to the concrete interpretation of
skeletons (natural semantics).

This manuel does not explain the internal representation of skeletal se-
mantics either. For further information about this topic, you can consult
The Necro Library gitlab repository.

To install Necro Lib, go to Necro Lib's git repository and follow the
instructions given by the README.md �le.

2 Skel: the language (natural interpretation)

2.1 Introduction

Skel is a minimalist strongly-typed functional language that allows to specify
the semantics (in particular the natural semantics) of a language (usually a
programming language). The Skel �les are called skeletal semantics.

Throughout this section, we will use the small-step semantics of λ-calculus
as an example.

A skeletal semantics is a set of declarations of two di�erent kinds, namely
type declarations and term declarations. Each declaration is either speci�ed

or unspeci�ed. The order of the declarations is irrelevant, and they are
assumed to be mutually dependant.

2.2 Type declarations

2.2.1 Built-in types

There are no built-in types in Skel. Though this can seem tedious, it is not
essential to describe a semantics, and providing built-in types would require
to choose implementations, which can be problematic. For instance, if we
choose to have a builtin string type, which allows all unicode characters, it
could not be used to describe Coq's semantics. Reciprocally, if we restrict the
string type to ascii characters, we cannot represent all of OCaml's strings.
And let's not even discuss the case of 32/31/64/. . . bits (or even arbitrary
precision) integers.

2.2.2 Unspeci�ed types

As we said, there are speci�ed and unspeci�ed types. For an unspeci�ed type,
we just give its name, to state that it exists. For instance, for λ-calculus, we
can choose not to specify how identi�ers are internally represented, since it
does not a�ect the semantics in itself. Therefore, we write:

type ident

The possibility to leave a type unspeci�ed is an important feature of Skel.
It is useful when we don't want to bother with internal representations, as
is the case with ident. It can also be used to leave aside the de�nition of a
term, in the context of a gradual implementation.

2.2.3 Product and arrow types

It is possible to construct product types. The product types are de�ned by
the syntax (t_1, ..., t_n). There is no product type with one component,
but the product type with 0 element (), also called unit type is a valid
product type. The elements of a product type are also written in the usual
way for a tuple. Therefore, if x has type t1 and y has type t2, then (x, y)

has type (t1, t2).
There are also arrow types, that is the type of functions. We construct

an arrow type using � or ->. For instance, if bool is the type of booleans,
then bool � bool is the type of functions that take a boolean as argument
and return a boolean.

It is possible to have arrow types with an arrow type on the right. It is
usual in some languages (e.g. OCaml) to curry functions as much as possible.
In Skel, we advise to be cautious, as uncurrying functions make for a clearer
application process, to only one argument at a time. To this end, we limit
application to one term at a time, as is explained below. For instance, when
working in an e�ectful world, the application f x y may cause e�ects either
when applying to x or to y, which is hidden by the multiple application.

2.2.4 Variant types

We can de�ne variant types, that is algebraic data types, de�ned by stating
the list of constructors, together with the type that each constructor expects.

type term =

| Var ident

| Lam (ident, term)

| App (term, term)

Here, we state that the type term has three constructors. The �rst
constructor, Var, represents a variable. It expects one argument, the variable
name. The second constructor Lam, represents a λ-abstraction. It expects
two arguments, or rather an argument, which is a pair. The �rst component
of this pair is the identi�er that is abstracted, and the second component is
the body of the abstraction. Finally, the third constructor App, represents
an application of one term to another, and thus expects a pair of terms as
argument.

It is important to notice that a constructor always expects exactly one
argument. In opposition to OCaml for instance, in which a constructor can
have 0, 1 or multiple arguments.

Therefore, if we want several arguments, we use product types. In the
same way, if we want no argument, we use the unit type, that is the product
type with no component.

type boolean =

| True ()

| False ()

This is inconvenient, as we don't want to carry an empty tuple every-
where. Skel o�ers a syntactic sugar that allows to omit the () in this context.
The following is syntactically equivalent to the preceding:

type boolean =

| True

| False

2.2.5 Record types

As in most programming languages, a record is a lot like a tuple with named
components. We de�ne them by stating the list of �elds, together with the
type of each �eld. We don't have any instance of it in the semantics of
λ-calculus so we give an example which does not relate to it.

type human = (name: string, size: int, birth: date)

Here, the type human has three components. If we have a term napoleon

of type human, we can access one of its components, e.g. its size, by using
the notation napoleon.size.

4



It is once again possible to de�ne recursive record types. Here is for
instance the de�nition of a stream type:

type stream = (head: int, tail: stream)

2.2.6 Type aliases

The last kind of type de�nition, is the de�nition of a type alias. It does
not de�ne a new type strictly speaking, but it creates an alias to refer to a
previously de�ned type. For this reason, aliases are not recursive. They can
refer to other aliases, but their is no circularity allowed.

type matrix22 = (pair_int, pair_int)

type pair_int = (int, int)

type binop = pair_int � int

2.2.7 Polymorphism

Types can be parametric. Polymorphism is explicitly declared. Below are
examples for each kind of type declaration.

For an unspeci�ed type, we give the de�nition, and we write _ for each
type argument since it is not used afterwards. To illustrate, we de�ne the
type of maps with two type arguments, being the type of keys, and the type
of values.

type map<_, _>

For a variant type, we give the type arguments when stating the type
name, and we can use them in the same way as normal types in the decla-
ration of the constructors' type. All constructors are then expected to take
the same type arguments as the type, in the same order. Here is for instance
the union type.

type union<a,b> =

| InjL a

| InjR b

For a record type, we give the type arguments when stating the type
name, and we can use them in the same way as normal types in the declara-
tion of the �elds' types. All �elds are then expected to take the same type
arguments as the type, in the same order. Here is for instance the pair type.

type pair<a,b> = (left: a, right: b)

Finally, type aliases can be polymorphic too. We de�ne them in the same
manner as the variant and record types. For instance, the option type can
be de�ned in the following way.

type option<a> = union<a,()>

2.3 Term declarations

Term declarations are the base element of the skeletal semantics. One can
also specify or not specify a term. If it is unspeci�ed, one only gives its name
and type. If it is speci�ed, one also gives an explicit term to de�ne it.

(* U n s p e c i f i e d *)

val assert: boolean � ()

val add: nat � nat � nat

(* S p e c i f i e d *)

val two: int = Succ (Succ Zero)

A term can be polymorphic. For instance, one can declare the term map

in the following way.

val map<a,b>: (a � b) � list<a> � list<b>

2.4 Terms and skeletons

First, we de�ne patterns. A pattern is either a wildcard, a variable matching
anything, a constructor applied to a pattern, a tuple of patterns, or a record
of patterns.

p ::= _

| x
| C p

| (p, . . . , p)
| (f = p, . . . , f = p)

Now let us de�ne terms and skeletons. Intuitively, the terms stand for
already computed expressions, and the skeletons stand for expressions that
still have to be evaluated.

A term can be:

� A variable

� A constructor applied to a term

� A tuple of terms

� A lambda-abstraction, which returns a skeleton (see below)

� A record

� A term (of record type) with some �eld rede�ned

� The access to a �eld of a term

� The access to a component of a term

t ::= x

| C t

| (t, . . . , t)
| λx : τ → S

| (f = t, . . . , f = t)

| t← (f = t, . . . , f = t)

| t.f
| t.n

Since Skel is a modern language, it uses the unicode characters λ, � and
�, but if you're less modern, or if your keyboard does not contain those
character, you may replace λ by \, � by -> and � by <-.

Note that the skeleton λp�s can also be written as λp·s.
As for the skeletons, there are one of the following:

� The immediate return of a term

� The application of a term to a term

� A branching, i.e. a list of zero, one or several skeletons

� A let-binding

� An existential quanti�er

� A pattern-matching on a term

S ::= t

| t t
| branchS . . . S end

| let p = S inS

| ∃τ
| match t with | p → S | . . . | p → S end

Once again, for those who prefer ascii character instead of unicode, you
can replace the existential with a question mark. That is, ∃τ and ?τ are
syntactically equivalent.

The following example shows an application for most of these possibili-
ties.

type ident

type term =

| Var ident

| Lam (ident, term)

| App (term, term)

val subst: (ident, term, term) � term

val ss (t:term): term =

match t with

| Var x -> branch end

| Lam (x, body) ->

let b' = ss body in

Lam (x, b')

| App (t1, t2) ->

branch

let t1' = ss t1 in

App (t1', t2)

or

let t2' = ss t2 in

App (t1, t2')

or (* ÿ -r e d u c t i o n *)

let Lam (x, body) = t1 in

subst (x, t2, body) (* b o d y [ x � t 2 ] *)

end

end

A given skeleton can be evaluated in a given environment to 0, 1 or
several results. One can understand the evaluation of a skeleton as a set, or
as a non-deterministic choice. We will choose the second point of view, that
is both closest to the computing approach of programming language and to
the inference rules of a natural semantics.

The evaluation of the let-binding let p = b in s happens in three parts.
First, b is evaluated to a value y. Then, we attempt to interpret this result
as being an instance of p. In particular, let Lam (x, body) = t1 in ...

means that t1 must be a lambda-abstraction in order for the skeleton to give
a result. Then, if we succeeded in matching the value v to the pattern p, the
free variables occurring in p are bound to the inferred values, and s is then
computed.

One can also use the notation s1 ; s2 which is a syntactic sugar for
let () = s1 in s2

The evaluation of a branching consists on a non-deterministic choice of
one of the branches that yield a result. For instance, the reduction of an
application may be a result of reducing the left term, the right term, or of a
beta-reduction, if the application happens to be a redex.

2.5 Binding Operators

When one wants to write a semantics using monads, to carry states, or
propagate exceptions for instance, we want to be able to describe those
monads in Skel. For this reason, Skel has a built-in system to handle monadic
binders.

Let us assume that we have a state monad m<·> and a term bind<a,b>

of type m<a> � (a � m<b>) � m<b>.
One can write

let p =%bind s1 in s2

The type arguments are inferred, and the result is semantically equivalent to
bind<a,b> s1 (λ p : τ � s2). It makes for a clearer, more concise, and
more faithful code.

It is also possible to map binders to symbols. For instance, if one
writes binder @s := bind, then one can bind using the following syntax:
let p =@s s1 in s2. We can also use the binding notation on semicolon.
s1 ;@s s2 is therefore a metanotation for let () =@s s1 in s2.

9



3 Appendices

3.1 Special comments

Comments beginning with (* * are called special comments. They are con-
sidered semantically relevant and can therefore not occur anywhere in a Skel
�le. A special comment ending on the line before a declaration is linked with
said declaration. A special comment beginning just after a constructor dec-
laration, on the same line, is linked to the constructor. A special comment
occuring in-between two declarations, or before or after any declaration, is
linked to the �le itself.

3.2 Splitting a semantics in several �les

Skel allows a way to split a semantics in several �les. It can be a way to
keep a functioning �le as a building block for several semantics, or a way to
have a clearer semantics, divided into several parts.

Every �le must be valid in itself. A �le begins by a list of require which
lists all semantics required to interpret the current Skel �le. They may them-
selves have other requires, but cycles are not allowed in the dependencies.
The rest of the contents consists of normal declarations as de�ned above

3.2.1 quali�ed identi�ers

The �les mainly act as name spaces. When one wants to access a type
(resp. a term, constructor, �eld) that was de�ned in a di�erent �le, there is
a speci�c syntax for that, a::x. This means that x was de�ned in �le a.sk.

3.3 BNF

In the BNF, presented in Figure 1, <ident> is an identi�er matching ei-
ther [a-z][a-zA-Z0-9_']* or _[a-zA-Z0-9_']+ and <Ident> is an iden-
ti�er matching [A-Z][a-zA-Z0-9_']*. A symbol is a special character,
followed by an arbitrary sequence of special characters or characters in
[a-zA-Z0-9_']. The accepted special characters are !&*⊤⊥@$^∀+-/~

To avoid considerations of priority, we consider the rules below as the
generated AST. But in reality, we use associativity, priorities and parentheses
to solve con�icts. For instance, "let ... in" takes priority over ";", and
"�" is right-associative.

sem ::= <decl>*

decl ::= type <ident>

| type <ident> = ( "|" <Ident> <type> )+

| type <ident> = ( <ident>:<type>, ..., <ident>:<type> )

| type <ident> = <type>

| term <ident>: <type>

| binder <symbol> = <ident>

| term <ident>: <type> = <term>

type ::= <ident>

| "(" <type> , ... , <type> ")"

| <type> � <type>

term ::= <ident>

| <Ident> <term>

| "(" <term> , ... , <term> ")"

| <term>.<ident>

| λ <ident> : <type> � <skeleton>

| "(" <ident> = <term> , ... , <ident> = <term> ")"

| <term> � "(" <ident> = <term> , ... , <ident> = <term> ")"

pattern ::= _

| <ident>

| <Ident> <pattern>

| "(" <pattern> , ... , <pattern> ")"

| "(" <ident> = <pattern> , ... , <ident> = <pattern> ")"

skeleton ::= <term>

| branch end

| branch <skeleton> or ... or <skeleton> end

| <term> <term>

| let <pattern> = <skeleton> in <skeleton>

| ∃ <type>

Listing 1: Skel's BNF (without polymorphism, special comments, or �les)

3.4 Reserved keywords

Necro's reserved keywords are the following (only lowercase):

binder branch end in let match or require type val with

